

V článku [Systémové proměnné a složky](#) jsme si povídali o tom, jak se dobrat speciálních složek operačního systému. Dnes se podíváme na to, jak složky a soubory vybírat prostřednictvím dialogů.

Excel nabízí jakýsi univerzální objekt FileDialog, který v sobě nese čtyři podtypy dialogů – pro výběr složky (Folder Picker), výběr souborů (File Picker), a pro otevření (Open), resp. uložení souboru (SaveAs).

Nejprve si ukážeme výběr složky.

```
1  Sub DialogVyberSlozky()  
2  
3      'víceúčelový dialog, zde pro výběr složky  
4      With Application.FileDialog(msoFileDialogFolderPicker)  
5  
6          'titulek, běžně "Procházet"  
7          .Title = "Výběr složky"  
8  
9          'výchozí styl zobrazení, zde velké ikony  
10         'Windows 7, 64 bit, nefunkční  
11         .InitialView = msoFileDialogViewLargeIcons  
12  
13         'vícenásobný výběr složek nelze použít  
14         '.AllowMultiSelect = True  
15  
16         'výchozí zobrazená složka, zde Temp  
17         .InitialFileName = Environ("Temp")  
18  
19         'popis tlačítka, běžně "OK"  
20         .ButtonName = "Vybrat"  
21  
22         'zobrazení dialogu  
23         .Show  
24  
25         'byla vybrána složka?  
26         If .SelectedItems.Count > 0 Then  
27  
28             'výpis do okna Immediate  
29             Debug.Print .SelectedItems(1)  
30  
31         End If  
32  
33     End With  
34  
35 End Sub
```



## Výběr složky

Na chvíli si odskočíme pro výběr složky do API a tak trochu se vrátíme k tématu systémových složek, abychom si nastavili výchozí složku v dialogu.

```

1  Private Type BROWSEINFO
2      hOwner As Long
3      pidlRoot As Long
4      pszDisplayName As String
5      lpszTitle As String
6      ulFlags As Long
7      lpfn As Long
8      lParam As Long
9      ilmage As Long
10 End Type
11
12 Private Const CSIDL_PERSONAL As Long = &H5
13
14 Private Declare Function SHGetPathFromIDLList Lib "shell32.dll" Alias _
15     "SHGetPathFromIDLListA" (ByVal pidl As Long, ByVal pszPath As String) As Long
16
17 Private Declare Function SHBrowseForFolder Lib "shell32.dll" Alias _
18     "SHBrowseForFolderA" (lpBrowseInfo As BROWSEINFO) As Long
19
20 Sub TestSlozkaVyber()
21
22     Dim strSlozka As String
23     strSlozka = GetDirectory()
24
25 End Sub
26
27 Function GetDirectory(Optional Msg) As String
28
29     Dim bInfo As BROWSEINFO
30     Dim path As String
31     Dim r As Long, x As Long, pos As Integer
32
33     'výchozí složka
34     'http://www.garybeene.com/code/visual%20basic252.htm
35     'zde moje složka Dokumenty
36     bInfo.pidlRoot = CSIDL_PERSONAL
37
38     'popisek dialogu
39     If IsMissing(Msg) Then
40         bInfo.lpszTitle = "Výběr složky"
41     Else
42         bInfo.lpszTitle = Msg
43     End If
44
45     'typ složky
46     bInfo.ulFlags = &H1
47
48     'zobrazení dialogu
49     x = SHBrowseForFolder(bInfo)
50
51     'parsování
52     path = Space$(512)
53     r = SHGetPathFromIDLList(ByVal x, ByVal path)
54     If r Then
55         pos = InStr(path, Chr$(0))
56         GetDirectory = Left(path, pos - 1)
57     Else
58         GetDirectory = ""
59     End If
60
61 End Function

```



## Výběr složky přes API funkce

Ale teď už zpět k objektu FileDialog a možnosti výběru souborů.

```
1      Sub DialogVyberSouboru()  
2  
3          Dim i As Integer  
4  
5          'víceúčelový dialog, zde pro výběr souborů  
6          With Application.FileDialog(msoFileDialogFilePicker)  
7  
8              'vícenásobný výběr souborů  
9              .AllowMultiSelect = True  
10  
11             'výchozí zobrazená složka, zde Temp  
12             .InitialFileName = Environ("Temp")  
13  
14             'zobrazení dialogu  
15             .Show  
16  
17             'byl vybrán nějaký soubor?  
18             If .SelectedItems.Count > 0 Then  
19  
20                 'pro každou vybranou položku  
21                 For i = 1 To .SelectedItems.Count  
22  
23                     'výpis do okna Immediate  
24                     Debug.Print .SelectedItems(i)  
25  
26                 Next i  
27  
28             End If  
29  
30         End With  
31  
32     End Sub
```

Určitě vás bude zajímat také možnost filtrování složky podle přípony souborů.

```

1  Sub DialogVyberSouboruFiltr()
2
3  Dim i As Integer
4
5  'víceúčelový dialog, zde pro výběr souborů
6  With Application.FileDialog(msoFileDialogFilePicker)
7
8      'vícenásobný výběr souborů
9      .AllowMultiSelect = True
10
11     'výchozí zobrazená složka, zde složka tohoto souboru
12     .InitialFileName = ThisWorkbook.path
13
14     'přidání dvou filtrů
15     .Filters.Add "Vybrané typy obrázků", _
16         "*.gif; *.jpg; *.jpeg; *.bmp; *.png", 1
17     .Filters.Add "Soubory aplikace Excel", "*.xl*", 2
18
19     'výchozí druhý filtr
20     .FilterIndex = 2
21
22     'zobrazení dialogu
23     .Show
24
25     'byl vybrán nějaký soubor?
26     If .SelectedItems.Count > 0 Then
27         'alternativně If .Show = -1 then
28
29         'pro každou vybranou položku
30         For i = 1 To .SelectedItems.Count
31
32             'výpis do okna Immediate
33             Debug.Print .SelectedItems(i)
34
35             Next i
36
37         End If
38
39     End With
40
41 End Sub

```



## Výběr souborů

Ještě se podíváme na třetí podtyp objektu FileDialog. Zatímco předchozí dialogy pouze vracely

seznam vybraných položek, tento je otevírá prostřednictvím metody Execute.

```
1  Sub DialogSouborOtevirit()
2
3  Dim i As Integer
4
5  'víceúčelový dialog
6  'zde pro výběr a otevření souborů
7  'a to pouze takových, které jsou přidružené
8  'hostitelské aplikaci, tj. Microsoft Excel!
9  With Application.FileDialog(msoFileDialogOpen)
10
11     'vícenásobný výběr souborů
12     .AllowMultiSelect = False
13
14     'výchozí zobrazená složka, zde Temp
15     .InitialFileName = ThisWorkbook.path
16
17     'zobrazení dialogu
18     .Show
19
20     'byl vybrán nějaký soubor?
21     If .SelectedItems.Count > 0 Then
22
23         'spuštění (lépe řečeno otevření souboru)
24         .Execute
25
26     End If
27
28 End With
29
30 End Sub
```

Jak už napovídá komentář v kódu, není bohužel možné si kupříkladu nastavit filtr na obrázky a čekat, že je metoda Execute spustí v přidruženém grafickém programu. Ostatně podívejte se na filtry tohoto dialogu.



Výběr souborů k otevření

I zde je cítit ze strany Microsoftu schizofrenie, vždyť už dávno existuje jiný způsob...

```

1  Sub DialogSouborOtevitAlternativa()
2
3  Dim varFName As Variant
4  Dim i As Integer
5
6  'metoda GetOpenFilename
7  varFName = _
8      Application.GetOpenFilename( _
9      FileFilter:="Soubory aplikace Excel (*.xl*), *.xl*", _
10     MultiSelect:=True)
11
12 If IsArray(varFName) Then
13     For i = LBound(varFName) To UBound(varFName)
14         'otevření sešitů
15         Workbooks.Open (varFName(i))
16     Next
17 End If
18
19 End Sub

```

Různorodost a změny v dialogích typu Otevřít/Uložit jsou leckdy přínosné (tipy si ukážeme jindy), ale nezřídka i pitomé a neergonomické (viz Excel 2013 a trotlovské proklikávání se k dialogu). Peklo s nimi zažívám ve smyslu kompatibility starších programů, které jednoduše spadnou, i když by jinak fungovali naprosto normálně. Obzvláště se to týká portable aplikací.